
1 Inleiding

Voor sommige functies van C-Gate is het nodig dat informatiewordt doorgegeven aanC-Gate vanuit het programmaCarola.
Raadpleeg de betreffende hoofdstukken voor de benodigde blokken en instructies voor de betreffende functionaliteit.

Als je de gegevensstructuur ReadVar of het commando OPCUAVAR in een andere context gebruikt, zorg er dan voor
dat de gebruikte gebieden elkaar niet overlappen. Houd er ook rekening mee dat de index die wordt gebruikt in het pro-
grammaCarola niet dezelfde index is opdewebinterface. DewaardenwordendoorgegevenaanC-Gate via dedatastructuur
ReadVar . In de controller is de eerste index één en de laatste 100. In C-Gate is de eerste index 0 en de laatste 99. Een
waarde die in de controller onder index 50 is opgeslagen, kan in C-Gate onder index 49 worden opgehaald.

Houd er rekening mee dat de Carola codevoorbeelden die hieronder worden uitgelegd mogelijk al gedeeltelijk of volledig
zijn opgeslagen in het masterprogramma. Vergelijk de secties en pas ze aan indien nodig. Duplicatie van functies en varia-
belen wordt door de Carola interpreter herkend als een fout.

Opmerking: Veranderingen in deQIROX WriteVar variabelen kunnen de volgorde van het Carola programmabeïnvloe-
den.

Opmerking: Gebruik Adobe Acrobat Reader om dit bestand te openen als u de Carola programmacode wilt kopiëren.

2 Totale efficiëntie van de apparatuur (OEE)

2.1 Bepaal de werkelijke hoeveelheid en de goede hoeveelheid

De volgende programmadelen moeten worden opgeslagen in respectievelijk het hoofdprogramma en het onderdelenpro-
gramma om de werkelijke hoeveelheid en de goede hoeveelheid te bepalen. Het is belangrijk dat de volgende procedures
uitsluitend het indexbereik 94 tot 100 gebruiken en niet voor andere doeleinden kunnen worden gebruikt. Het bereik kan
in de toekomst groter worden.

2.1.1 Masteropleiding

De eerste regels zijn nodig omde variabelen vanuit andere programma’s te kunnen importeren en schrijven. Ze staan direct
na de instructie RESTART .

RESTART
! VARIABELE VERKLARING
VAR SEL, ERG
STRING VAR COMPSTAT, COMPQUAL
PUBLIC STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM

Dit blok definieert eenprocedure die informatie stuurt naar C-Gate dat een component start. Hiervoormoetendehieronder
gebruikte variabelen worden ingesteld in het programma van de component.

PUBLIC PROC COMPSTART
COMPSTAT := ''
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
WAITM(100)

STRHND(2, COMPTYPE, COMPTYPE, ERG)
IF ERG=0 THEN COMPTYPE:='NODATA'
STRHND (0,COMPTYPE,'NODATA',ERG)
IF ERG=1 THEN BEGIN
LOPCUA1:
WRITE ('VARIABELE >COMPTYPE< NOG NIET BESCHREVEN!')
WAITM(500)
PAUSE
JUMP LOPCUA1
END

COMPSTAT := 'START'
COMPQUAL := 'IN_PROGRESS'
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99, COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI,'COMPSERNO')
OPCUAVAR(0,97, COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
ENDP

Dit blok definieert een procedure die doorgeeft aan C-Gate dat een component klaar is.

! EINDE VAN COMPONENTOPNAME
PUBLIC PROC COMPEND
COMPSEAM := ''
COMPSTAT := 'END'
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
WAITM(100)

COMPTYPE := ''
COMPVARI := ''
COMPSERI := ''
COMPSTEP := ''
COMPSTAT := ''
COMPQUAL := ''
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99, COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI,'COMPSERNO')
OPCUAVAR(0,97, COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL,'COMPQUALITY')

OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

Dit blok definieert een functie die de informatie doorgeeft aan C-Gate dat er een nieuwe naad wordt gestart. De naam van
de naadmoet in de variabele COMPSEAM worden geschreven voordat de procedure wordt aangeroepen.

! EEN NAAD BEGINNEN
PUBLIC PROC SEAMSTRT
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

Dit blok definieert een functie die de informatie doorgeeft aan C-Gate dat de huidige naad klaar is.

! EINDE VAN EEN NAAD
PUBLIC PROC SEAMEND
COMPSEAM := ''
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

Dit blok definieert een functie die opdePHGopvraagt of het onderdeelOK is of niet. De controle is gebaseerd op een visuele
inspectie door de werker en blokkeert verdere programmastromen totdat er een invoer is gedaan. Je kunt de formulering
IS DE COMPONENT OK=1 OF NIET OK=0 ? aanpassen aan je vereisten.

! OPNAME VAN KWALITEITSONDERDELEN
PUBLIC PROC COMPQMAN
WREAD ('IS DE COMPONENT OK=1 OF NIET OK=0 ?', SEL )
IF SEL=1 THEN COMPQUAL := 'OK' ELSE COMPQUAL := 'NOT_OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP

Dit blok definieert een functie die automatisch (zonder query) de kwaliteit van het onderdeel op OK zet. Deze functie kan
gebruikt worden als je de kwaliteit van de overall equipment effectiveness (OEE) niet dynamischwilt registreren,maar toch
de hele metric correct wilt laten berekenen.

! OPNAME VAN KWALITEITSONDERDELEN
PUBLIC PROC COMPQAUT
COMPQUAL := 'OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP

2.1.2 Variabelen

De volgende variabelen kunnen in het programmaworden gebruikt.

• COMPTYPE Type onderdeel.

• COMPVARI : Variant van de component.

• COMPSERI Serienummer van de component. Als er geen waarde is ingesteld, maakt C-Gate automatisch een serie-
nummer aan op basis van een tijdstempel.

• COMPSTEP Naam van de productiestap. De waarde is belangrijk als het onderdeel in verschillende stations wordt
verwerkt.

• COMPSEAM Naam van de naad. Een naad kan uit verschillende websegmenten bestaan.

2.1.3 Eenvoudig onderdelenprogramma

Importeer de variabelen en procedures uit het masterprogramma.

RESTART

! IMPORTEREN VAN VARIABELEN EN PROCEDURES
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER

MAIN

Stel in deprogrammastroom tenminste de verplichte velden in en, indien nodig, ookdeoptionele velden. Zodra alle velden
zijn ingesteld, roept u de procedure COMPSTART op. Als je de module C-Gate.QM hebt gekocht, kun je bovendien naden
definiëren en het componentrecord fijner structureren. Gebruik hiervoor de procedures SEAMSTRT en SEAMEND . Zo-
dra het werkstuk klaar is en je de werkstukkwaliteit wilt instellen vanuit het programma Carola, roep je COMPQAUT of
COMPQMAN op om de kwaliteitsgegevens over te dragen. Als je de module C-Gate.QM hebt gekocht, wordt de onderdeel-
kwaliteit automatisch bepaald met behulp van de QIROX SD bewaking. Het oproepen van de procedure COMPQAUT of
COMPQMAN heeft geen effect op het onderdeelprotocol.

Om de component te beëindigen, roep je de procedure COMPEND aan. De volgorde van de aanroepen is belangrijk.

Houd er rekening mee dat er maximaal een seconde moet verstrijken tussen het aanroepen van COMPEND en het her-
starten van een component met COMPSTART , anders kan het eindsignaal verloren gaan door een datasynchronisatie in
de controller. Het is de bedoeling dat deze beperking wordt verwijderd in een van de komende softwareversies voor de
robotbesturing.

! INSTELLEN VAN DE COMPONENTINFORMATIE
! VERPLICHT VELD
COMPTYPE := 'SCHAKELKAST'

! OPTIONEEL
! COMPVARI := 'XL'

! OPTIONEEL
! COMPSERI := '1234567890'

! OPTIONEEL
! COMPSTEP := 'VLOER'

! BEGIN VAN DE COMPONENT
CALL COMPSTART

! ***************
! ONDERDEEL PROGRAMMA
! ***************

COMPSEAM := 'VULLAS 1'
CALL SEAMSTRT

! ***************
! PROGRAMMA VULLAS 1. BIJVOORBEELD VERSCHILLENDE GC-BEWEGINGEN.
! ***************

COMPSEAM := 'VULLAS 2'
CALL SEAMSTRT

! ***************
! PROGRAMMA VULLAS 2. BIJVOORBEELD VERSCHILLENDE GC-BEWEGINGEN.
! ***************

CALL SEAMEND

! STEL DE KWALITEIT VAN HET ONDERDEEL IN HET PROGRAMMA CAROLA IN.
! BEL VOOR COMPEND
CALL COMPQMAN
! OF
! CALL COMPQAUT

! EINDE VAN HET ONDERDEEL
! BEL VOOR COMPQMAN OF COMPQAUT
CALL COMPEND

! EINDE VAN HET DEELPROGRAMMA
END

2.1.4 Complex onderdelenprogramma

Het volgende Carola programma maakt een onderdeel met twee productiestappen. De productiestappen kunnen achter-
eenvolgens gelast worden op één robot of parallel op meerdere robots. Voorwaarde hiervoor is dat de stappen worden
gecombineerd in een protocol:

• Type, variant en serienummer moeten op alle robots hetzelfde zijn ingesteld.

• In de overeenkomstige componentensjabloon moet het juiste minimumaantal productiestappen worden ingesteld.
In dit geval worden er twee productiestappen gelast, dus moet het aantal worden ingesteld op minstens de waarde
twee.

RESTART
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
MAIN

! ***************
! PRODUCTIESTAP 1
! ***************

COMPTYPE := 'KETEL'
COMPVARI := '80KW'
COMPSERI := '0042012342'
COMPSTEP := 'VLOER'

CALL COMPSTART

COMPSEAM := 'VULLAS 1'
CALL SEAMSTRT
! PROGRAMMA VULLAS 1. BIJVOORBEELD VERSCHILLENDE GC-BEWEGINGEN.

COMPSEAM := 'VULLAS 2'
CALL SEAMSTRT
! PROGRAMMA VULLAS 2. BIJVOORBEELD VERSCHILLENDE GC-BEWEGINGEN.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

! ***************
! PRODUCTIESTAP 2
! ***************

COMPTYPE := 'KETEL'
COMPVARI := '80KW'

COMPSERI := '0042012342'
COMPSTEP := 'DEKKING'

CALL COMPSTART

COMPSEAM := 'VULLAS 3'
CALL SEAMSTRT
! PROGRAMMA VULLAS 3. BIJVOORBEELD VERSCHILLENDE GC-BEWEGINGEN.

COMPSEAM := 'VULLAS 4'
CALL SEAMSTRT
! PROGRAMMA VULLAS 4. BIJVOORBEELD VERSCHILLENDE GC-BEWEGINGEN.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

END

3 Opmerkingen

Als u een PDM- of UMS-klant bent of was en een van de twee opties is geactiveerd in de configuratie van de regelaar, moet
u dringend de volgende opmerking in acht nemen! Als een van de twee opties is geactiveerd, maar de robotregelaar niet
is aangesloten op de bijbehorende pc-toepassing, zullen er gaten ontstaan in de gegevensstroom die tot 5 seconden lang
kunnen zijn en die het componentenprotocol aanzienlijk vervalsen. Sluit de robotbesturing aan op de pc-toepassing of
verwijder de betreffende optie uit de configuratie.

Als u met meerdere robots tegelijk aan hetzelfde onderdeel werkt en alle gegevens in één onderdeelprotocol binnen de
module C-Gate.QMwilt samenvoegen, neemdan contact opmet CLOOSService voormeer informatie. Laat anders het veld
COMPSERI leeg, omdat dit kan leiden tot ongewenste neveneffecten bij de berekening van de totale effectiviteit van de
apparatuur (OEE).

4 Variabelen

Je hebt de mogelijkheid om variabelen toe te voegen aan het componentprotocol vanuit het programma Carola. Gebruik
hiervoor deopdrachtOPCUAVAR (zie voorbeeld). Als denaam (laatste parameter) het voorvoegsel ’COMPVAR_’ heeft,wordt
de variabele aan de component gekoppeld. Als de naam het voorvoegsel ’STEPVAR_’ heeft, wordt de variabele gekoppeld
aande componentstap. Als de naamhet voorvoegsel ’SEAMVAR_’ heeft,wordt de variabele gekoppeld aandehuidige naad.
Voor elke geselecteerde naam kan slechts één waarde worden opgeslagen. De laatste waarde die het laatst werd ingesteld
bij de respectieve naam,wordt opgeslagen in het logboek. Het voorvoegsel vandenaamwordt niet opgeslagen. Dewaarde
moet opnieuwworden ingesteld voordat de naad, stap of componentwordt voltooid. Als dit niet gebeurt, wordt dewaarde
ook toegevoegd aan het volgende element. Gehele getallen, kommagetallen en teksten kunnen worden opgeslagen.

CALL SEAMSTRT

OPCUAVAR(0,1, 'EXAMPLE-VALUE','COMPVAR_SEAM_MEASURE_A')

OPCUAVAR(0,1, '','')
CALL SEAMEND

